
SAS: Interfacing PROC MIANALYZE with PROC PANEL 

 

As shown in Figure 1, the dataset raw_input contains a cross-sectional variable “country”, a time 
variable “year”, a response “Y”, and two covariates named “X1”, “X2”. 

 

 

Figure 1. Raw panel dataset 

Since covariates contain missing values, we have to impute them first by applying MCMC in 
PROC MI: 

 
proc mi data = raw_input seed = 42037921 nimpute = 100  

   out = imputed_data; 
 mcmc timeplot(mean(x1) mean(x2)); 
 var x1 x2; 
 
run; 
 

which gives us 100 imputed datasets contained in a single dataset imputed_data (Figure 2). 

 



 

Figure 2. Dataset imputed_data obtained from  PROC  MI. 

 

Our final goal is to regress Y on X1, X2, and the first lag of Y. The latter is created through the 
following code: 

proc sort data = imputed_data; 
 by _Imputation_ country year; 
run; 
 
proc panel data = imputed_data; 
 by _Imputation_; 
 id country year; 
 clag Y(1) / out = imputed_data; 
run; 
 

First, we fit a model in PROC REG, because MIANALYZE can be interfaced seamlessly with 
REG as follows: 

 



proc reg data = imputed_data outest = out_reg covout noprint; 
   model Y = Y_1 X1 X2; 
   by _Imputation_;      
run; 
quit; 
 
proc mianalyze data = out_reg; 
 modeleffects Intercept Y_1 X1 X2;       
run; 
 
 
We need to do a similar thing with PROC PANEL. One of the reasons is that PANEL can do 
GMM estimation which is more appropriate than OLS when a lagged response is present in the 
right-hand side.  We have to make the following code work: 

proc panel data = imputed_data outest = out_pan covout noprint; 
 by _Imputation_;  
 id country year; 
 inst  constant depvar exogenous = (X1 X2); 
 model Y = Y_1 X1 X2 / itgmm nolevels;  
run; 
 
 
/* Here some code will be inserted   (A) */ 
 
proc mianalyze data = out_pan; 
 modeleffects Intercept Y_1 X1 X2;       
run; 
 
 

Unfortunately, this code fails without some additional statements in line (A). While the structure 
of out_pan is similar to that of out_reg (Figure 3), PROC MIANALYZE cannot take out_pan “as 
is”. 



 

Figure 3. Output of PROC REG, out_reg (top) and output of PROC PANEL, out_pan (bottom). 

 

For instance, the output variable _TYPE_  is supposed to tell apart point estimates of regression 
coefficients from their variance-covariance matrix. For point estimates, _TYPE_ should be 
“PARM” or “PARMS”, but in out_pan it is equal to “ITGMM”. Another distinction is that 
PANEL outputs MSE as opposed root MSE in REG.  

Taking information from out_pan and converting it manually into out_reg’s format seems like a 
way to go. The catch is that it will fail unless you also assign correct TYPE of the dataset (do not 
confuse a dataset attribute TYPE with the variable _TYPE_ mentioned above). The TYPE 
attribute of out_reg is equal to EST. If you create a copy of out_reg as: 

 
data out_reg2; 
 set out_reg; 
run; 



 

then, unexpectedly, PROC MIANALYZE won’t accept out_reg2: 

47   proc mianalyze data = out_reg2; 
48        modeleffects Intercept Y_1 X1 X2; 
49   run; 
 
ERROR: The input TYPE= data set is not a valid data set without 
specifying variables for standard errors in the STDERR statement. 
 

because the TYPE attribute is not copied from out_reg to out_reg2. After this is taken into 
account, out_pan becomes good enough for MIANALYZE and the final code looks as follows: 

 

proc panel data = imputed_data outest = out_pan covout noprint; 
 by _Imputation_;  
 id country year; 
 inst  constant depvar exogenous = (X1 X2); 
 model Y = Y_1 X1 X2 / itgmm nolevels;  
run; 
 
                         /* Code (A) begins */ 
 
/* Compute _RMSE_ : */ 
data out_pan; 
 set out_pan; 
 retain _RMSE_; 
 if (_MSE_ ne .) then _RMSE_ = sqrt(_MSE_); 
 output; 
run; 
 
/* Re-assign some values and drop redundant columns: */ 
data out_pan;  
 set out_pan; 
 if _TYPE_ eq 'ITGMM' then _TYPE_ = 'PARMS'; 
 if _TYPE_ eq 'COVB' then _TYPE_ = 'COV'; 
 _MODEL_ = 'MODEL1'; 
 drop _METHOD_  _CSID_   _MSE_;  
run; 
 
 
 
 
 



/* Change the order of variables */ 
data out_pan; 
 RETAIN _Imputation_ _MODEL_ _TYPE_  
    _NAME_ _DEPVAR_ _RMSE_ 
    Intercept  
    Y_1 X1 X2 Y;   
 set out_pan; 
run; 
 
/* Set outpan's type to EST */ 
proc datasets; 
 modify out_pan (TYPE = EST); 
quit; 
 
                      /* Code (A) ends */ 
 
proc mianalyze data = out_pan; 
 modeleffects Intercept Y_1 X1 X2;       
run; 
 

 


